Задатак

ТТ реализација филтра (Tow-Thomas state-variable biquad, MAX274 active filter) има познате вредности елемената $R_1 = \frac{3}{2} R$, $R_2 = R_3 = R$, $R_4 = \frac{3}{4} R$, $R_7 = R_8 = R$, $C_1 = C_2 = \frac{1}{R\,\Omega}$ и познати су реални параметри $R, \Omega > 0$.
Одредити трансфер функцију $\underline{H}(\underline{s}) = \frac{V_3(\underline{s})}{\underline{U}_g(\underline{s})}$ (уопштену преносну комплексну функцију електричног кола у области унилатералне Лапласове трансформације, трансмитансу напона), њене нуле и полове.
Одредити амплитудски одзив $M(\omega)$, пропусни опсег 3dB, његову ширину и граничне учестаности.
Нацртати амплитудску карактеристику. Обележити осе графика, координатни почетак, пресеке и додире графика са осама, и тачке екстремума.
Одредити импулсни одзив (Гринову функцију) и одскочни одзив (индициону функцију) за напон чвора 3, као и њихов домен (област дефинисаности у времену).



Решење

Анализа помоћу унилатералне Лапласове трансформације је делотворан и учинковит поступак за решавање линеарних временски-непроменљивих електричних кола. Користићемо Лапласову трансформацију за одређивање одзива и функција кола.

Директна унилатерална Лапласова трансформација је дефинисана као
$\underline{U}(\underline{s}) = \mathrm{LT}(u(t)) = \int_{0^-}^{\infty} u(t)\, e^{-\underline{s}\,t}\, dt$, где је $\underline{s}$ Лапласова променљива, коју још називамо комплексна учестаност, а $(u(t), \underline{U}(\underline{s}))$ чини Лапласов трансформациони пар. Функција времена $u(t)$ је тренутна вредност (оригинал, напон/струја) и то је каузална функција времена. Комплексна функција $\underline{U}(\underline{s})$ је Лапласов трансформат (слика, комплексан напон/струја, комплексан представник, трансформат).

Посматрајмо линеарно временски непроменљиво електрично коло без почетне енергије у коме делује само један извор (један независан генератор). Сви природни почетни услови у тренутку нула-минус су једнаки нули. Уопштена комплексна функција електричног кола (функција електричне мреже, функција система) је однос трансформата одзива и трансформата побуде. Ако су побуда и одзив на истом приступу, комплексна функција се назива улазна комплексна функција (улазна имитанса). Ако су побуда и одзив на различитим приступима, комплексна функција се назива преносна комплексна функција (трансфер функција, комплексна функција преноса у области унилатералне Лапласове трансформције).



Користићемо сажети уопштени поступак напона чворова (CMNA, Compacted Modified Nodal Analysis) за постављање комплексних једначина кола. За потенцијал чвора 1 нећемо уводити посебну променљиву, већ ћемо користити једначину елемента $\underline{V}_1(\underline{s}) = \underline{U}_g(\underline{s})$. За потенцијале чворова 5, 6 и 7 нећемо уводити посебне променљиве, већ ћемо користити једначине елемента $\underline{V}_5(\underline{s}) = 0$, $\underline{V}_6(\underline{s}) = 0$, $\underline{V}_7(\underline{s}) = 0$. За улазне струје операционих појачавача нећемо уводити посебне променљиве, већ ћемо користити једначине елемента и сматрати да су те струје једнаке нули. У задатку се не траже струја напонског извора и излазне струје операционих појачавача, тако да можемо изоставити једначине у којима се те струје појављују, а то су једначине за чворове 1, 2, 3 и 4.

Комплексне једначине CMNA електричног кола се постављају за чворове 5, 6 и 7, а променљиве су комплексни потенцијали чворова 2, 3 и 4.

$$\frac{0-\underline{U}_g(\underline{s})}{R_4} + C_1\,\underline{s}(0-\underline{V}_2(\underline{s})) + \frac{0-\underline{V}_2(\underline{s})}{R_1} + \frac{0-\underline{V}_4(\underline{s})}{R_3} + 0 = 0$$

$$\frac{0-\underline{V}_2(\underline{s})}{R_7} + \frac{0-\underline{V}_3(\underline{s})}{R_8} + 0 = 0$$

$$\frac{0-\underline{V}_3(\underline{s})}{R_2} + C_2\,\underline{s}(0-\underline{V}_4(\underline{s})) + 0 = 0$$

Заменом задатих вредности елемената, $R_1 = \frac{3}{2}R$, $R_2 = R_3 = R$, $R_4 = \frac{3}{4}R$, $R_7 = R_8 = R$, $C_1 = C_2 = \frac{1}{R\,\Omega}$, систем једначина кола се своди на

$$-\frac{\underline{U}_g(\underline{s})}{\frac{3}{4}R} - \frac{1}{R\,\Omega}\,\underline{s}\,\underline{V}_2(\underline{s}) - \frac{\underline{V}_2(\underline{s})}{\frac{3}{2}R} - \frac{\underline{V}_4(\underline{s})}{R} = 0$$

$$-\frac{\underline{V}_2(\underline{s})}{R} - \frac{\underline{V}_3(\underline{s})}{R} = 0$$

$$-\frac{\underline{V}_3(\underline{s})}{R} - \frac{1}{R\,\Omega}\,\underline{s}\,\underline{V}_4(\underline{s}) = 0$$

Природни почетни услови се не појављују у једначинама зато што се комплексна функција дефинише за коло без почетне енергије.

$$\frac{4}{3}\,\underline{U}_g(\underline{s}) + \frac{1}{\Omega}\,\underline{s}\,\underline{V}_2(\underline{s}) + \frac{2}{3}\,\underline{V}_2(\underline{s}) + \underline{V}_4(\underline{s}) = 0$$

$$\underline{V}_2(\underline{s}) + \underline{V}_3(\underline{s}) = 0$$

$$\underline{V}_3(\underline{s}) + \frac{1}{\Omega}\,\underline{s}\,\underline{V}_4(\underline{s}) = 0$$

Ради боље контроле димензија једначина и израза подесно је увести смену $\underline{p} = \frac{1}{\Omega}\,\underline{s}$, која уводи бездимензиону нормализовану комплексну учестаност $\underline{p}$ (чист број).

$$4\,\underline{U}_g(\underline{s}) + 3\,\underline{p}\,\underline{V}_2(\underline{s}) + 2\,\underline{V}_2(\underline{s}) + 3\,\underline{V}_4(\underline{s}) = 0$$

$$\underline{V}_2(\underline{s}) + \underline{V}_3(\underline{s}) = 0$$
$$\underline{V}_3(\underline{s}) + \underline{p}\,\underline{V}_4(\underline{s}) = 0$$

Из последње две једначине можемо изразити $\underline{V}_2(\underline{s})$ и $\underline{V}_4(\underline{s})$ у функцији $\underline{V}_3(\underline{s})$ и заменити у прву једначину.

$$4\,\underline{U}_g(\underline{s}) - 3\,\underline{p}\,\underline{V}_3(\underline{s}) - 2\,\underline{V}_3(\underline{s}) - 3\,\frac{1}{\underline{p}}\,\underline{V}_3(\underline{s}) = 0$$

Сређивањем се добија

$$4\,\underline{p}\,\underline{U}_g(\underline{s}) = \left(3\,\underline{p}^2 + 2\,\underline{p} + 3\right)\underline{V}_3(\underline{s})$$

Тражена трансфер функција је

$$\underline{H}(\underline{s}) = \frac{\underline{V}_3(\underline{s})}{\underline{U}_g(\underline{s})} = \frac{4\,\underline{p}}{3\,\underline{p}^2+2\,\underline{p}+3} = \frac{4\,\Omega\,\underline{s}}{3\,\underline{s}^2+2\,\Omega\,\underline{s}+3\,\Omega^2}$$

Уопштена комплексна функција електричног кола са концентрисаним елементима је рационална функција по комплексној учестаности $\underline{s}$ и са реалним коефицијентима. Трансфер функција се може представити као количник два полинома по $\underline{s}$.

$$H(\underline{s}) = \frac{\underline{B}(\underline{s})}{\underline{A}(\underline{s})},\ \underline{A}(\underline{s}) = 3\,\underline{s}^2 + 2\,\Omega\,\underline{s} + 3\,\Omega^2,\ \underline{B}(\underline{s}) = 4\,\Omega\,\underline{s}$$

Нуле комплексне функције електричног кола су корени (нуле) бројитеља.

$$\underline{B}(\underline{s}) = 0,\ 4\,\Omega\,\underline{s} = 0,\ \underline{s}_{z1} = 0$$

Полови комплексне функције електричног кола су корени (нуле) именитеља.

$$\underline{A}(\underline{s}) = 0,\ 3\,\underline{s}^2 + 2\,\Omega\,\underline{s} + 3\,\Omega^2 = 0,\ \underline{s}_{p1} = \Omega\,\frac{-1-\mathrm{j}2\sqrt{2}}{3},\ \underline{s}_{p2} = \Omega\,\frac{-1+\mathrm{j}2\sqrt{2}}{3}$$

Комплексна функција електричног кола се може представити преко нула и полова (pole-zero-gain form).

$$\underline{H}(\underline{s}) = K\,\frac{(\underline{s}-\underline{s}_{z1})}{\left(\underline{s}-\underline{s}_{p1}\right)\left(\underline{s}-\underline{s}_{p2}\right)},\ K = \frac{4}{3}\,\Omega$$

Комплексна функција електричног кола има реалне коефицијенте зато што су коефицијенти у дефиниционим једначинама електричних елемената реални. Нуле и полови су зато или реални или у комплексно-конјугованим паровима.

Уобичајено је да се квадратни трином који одговара комплексно-конјугованом пару нула или полова представља као

$$\underline{A}(\underline{s}) = a_2\,\underline{s}^2 + a_1\,\underline{s} + a_0,\ \underline{A}(\underline{s}) = a_2\left(\underline{s}^2 + \frac{\omega_p}{Q_p}\,\underline{s} + \omega_p^2\right)$$

где је ω_p потег пола а Q_p сачинитељ доброте или Q-фактор пола.

Фреквенцијски одзив је комплексна функција електричног кола на имагинарној оси, $\underline{s} = \mathrm{j}\,\omega$.
Амплитудски одзив је модул фреквенцијског одзива. Фазни одзив је аргумент фреквенцијског одзива. Групно кашњење је негативан извод фазног одзива по угаоној учестаности.

$$M(\omega) = |\,\underline{H}(\mathrm{j}\,\omega)\,| = \left|\,\frac{4\,\Omega\,(\mathrm{j}\,\omega)}{3\,(\mathrm{j}\,\omega)^2+2\,\Omega\,(\mathrm{j}\,\omega)+3\,\Omega^2}\,\right|,$$

$$M(\omega) = \frac{4\,\Omega\,|\omega|}{|-3\,\omega^2+\mathrm{j}\,2\,\Omega\,\omega+3\,\Omega^2|} = \frac{4\,\Omega\,|\omega|}{\sqrt{\left(3\,\Omega^2-3\,\omega^2\right)^2+(2\,\Omega\,\omega)^2}}$$

$$M(\omega) = \frac{4\,\Omega\,|\omega|}{\sqrt{9\,\omega^4 - 14\,(\Omega\,\omega)^2 + 9\,\Omega^4}}$$

Фреквенцијске карактеристике су графици групног кашњења, амплитудског и фазног одзива у функцији угаоне учестаности [rad/s] или учестаности [Hz]. Фреквенцијске карактеристике се обично цртају у подесној логаритамској размери.

In[1]:=
```
Plot[4 p/Sqrt[9 p^4 - 14 p^2 + 9], {p, 0, 7}, PlotRange → {-0.2, 2.2},
 AxesLabel → {ω/Ω, M[ω]}, PlotLabel → "Амплитудска карактеристика",
 Epilog → {Text["(0,0)", {0.3, -0.1}], Text["(1,2)", {1.3, 2.1}],
   Text["M → 0+", {6.5, 0.1}]}, GridLines → Automatic]
```

Out[1]=



Због подеснијег приказа амплитудске карактеристике уводи се логаритамска размера по ординати. Децибел (dB) је двадесет декадних логаритама од посматране величине која је бездимензиона (чист број).

In[2]:=
```
Plot[20 Log[10, 4 p/Sqrt[9 p^4 - 14 p^2 + 9]], {p, 10^-1, 7}, PlotRange → All,
 AxesLabel → {ω/Ω, M_dB[ω]}, PlotLabel → "Амплитудска карактеристика [dB]",
 Epilog → Text["(0,0)", {0.3, 0.8}], GridLines → Automatic]
```

Out[2]=



Амплитудска карактеристика у максимуму има вредност од приближно 6 dB.

In[3]:= **20 Log[10, 2.0]**

Out[3]= 6.0206

Пропусни опсег је скуп учестаности на којима је квадрат амплитудске карактеристике већи од половине квадрата вредности на упоредној учестаности.
Пропусни опсег се изражава и преко учестаности, f, у херцима (Hz) и преко угаоне (кружне) учестаности, ω, у радијанима по секунди (rad/s).

Упоредна учестаност је учестаност максимума амплитудске карактеристике када је она једноставна функција учестаности, монотоно опадајућа, монотоно растућа, или звонаста крива са једним израженим локалним максимумом. Овако дефинисан пропусни опсег се зове пропусни опсег 3 dB.

Одредимо прво локални максимум амплитудске карактеристике. Због математичког сређивања израза подесније је одредити локални минимум квадрата реципрочне вредности.

$$M(\omega) = \frac{4\,\Omega\,|\omega|}{\sqrt{9\,\omega^4 - 14\,(\Omega\,\omega)^2 + 9\,\Omega^4}},$$

$$F(\omega) = \frac{1}{M(\omega)^2} = \frac{9\,\omega^4 - 14\,(\Omega\,\omega)^2 + 9\,\Omega^4}{16\,\Omega^2\,\omega^2},\ \ F(\omega) = \frac{9\,\omega^2}{16\,\Omega^2} - \frac{7}{8} + \frac{9\,\Omega^2}{16\,\omega^2}$$

$$\frac{dF(\omega)}{d\omega} = \frac{9}{16\,\Omega^2}\,2\,\omega - 2\,\frac{9\,\Omega^2}{16\,\omega^3},\ \ \frac{dF(\omega)}{d\omega} = 0,$$

$$\frac{9}{16\,\Omega^2}\,2\,\omega - 2\,\frac{9\,\Omega^2}{16\,\omega^3} = 0,\ \ \omega_{\mathrm{min},F(\omega)} = \Omega,\ \ \omega_{\mathrm{max},M(\omega)} = \Omega$$

$$\omega_{\mathrm{ref}} = \omega_{\mathrm{max},M(\omega)} = \Omega,\ \ M_{\mathrm{ref}} = M(\omega_{\mathrm{ref}}) = 2$$

Пропусни опсег је скуп учестаности $\omega_1 \le \omega \le \omega_2$ за које важи $M(\omega)^2 \ge \frac{1}{2}\,M_{\mathrm{ref}}^2$. Алтернативно, услов се исказује и као $M(\omega) \ge \frac{1}{\sqrt{2}}\,M_{\mathrm{ref}}$.

Физичко тумачење услова за пропусни опсег је следеће: Снага излазног сигнала у пропусном опсегу је већа или једнака од половине снаге улазног сигнала.

$$M(\omega)^2 \ge \frac{1}{2}\,M_{\mathrm{ref}}^2,\ \ \frac{16\,(\Omega\,\omega)^2}{9\,\omega^4 - 14\,(\Omega\,\omega)^2 + 9\,\Omega^4} \ge \frac{1}{2}\,2^2$$

$$\frac{8\,(\Omega\,\omega)^2}{9\,\omega^4 - 14\,(\Omega\,\omega)^2 + 9\,\Omega^4} = 1,\ \ p = \frac{\omega}{\Omega} \ge 0,\ \ 9\,p^4 - 22\,p^2 + 9 = 0,$$

$$p^2 = \frac{22 \pm \sqrt{22^2 - 4\cdot 81}}{18},\ \ p^2 = \frac{22 \pm 2\,\sqrt{121 - 81}}{18}$$

$$p^2 = \frac{11 \pm 2\,\sqrt{10}}{9},\ \ p^2 = \frac{10 \pm 2\,\sqrt{10} + 1}{9},\ \ p^2 = \frac{\left(\sqrt{10} \pm 1\right)^2}{3^2},\ \ p = \frac{\sqrt{10} \pm 1}{3}$$

Граничне учестаности пропусног опсега су реалне позитивне величине.

$$\omega_1 = \frac{\sqrt{10} - 1}{3}\,\Omega,\ \ \omega_2 = \frac{\sqrt{10} + 1}{3}\,\Omega$$

Ширина пропусног опсега је разлика граничних учестаности и то је реална позитивна величина. Може бити и бесконачна.

$$B_{\omega,3\ \mathrm{dB}} = \omega_2 - \omega_1,\ \ B_{\omega,3\ \mathrm{dB}} = \frac{2}{3}\,\Omega$$

Ако је пропусни опсег само један интервал по оси учестаности, дефинише се јединствена доња гранична учестаност (lower frequency edge/cutoff), јединствена горња гранична учестаност (upper frequency edge/cutoff) и ширина пропусног опсега (frequency bandwidth) као разлика горње и доње граничне учестаности. Доња гранична учестаност може бити једнака нули, а горња гранична учестаност може бити у бесконачности.

У овом задатку је $0 < \omega_1 < \omega_2 < +\infty$, тако да се анализирано електрично коло зове филтар пропусник учестаности (bandpass filter, BP filter, BPF).

Постоје и друге дефиниције пропусног опсега. Пажљиво прочитати документацију електричне компоненте, склопа, опреме,…, да би утврдили како је дефинисан и мерен пропусни опсег.

Импулсни одзив (Гринова функција, impulse response, unit impulse response) је инверзна унилатерална Лапласова трансформација одговарајуће трансфер функције (уопштене комплексне функције електричног кола у области Лапласове трансформације).

Инверзну унилатералну Лапласову трансформацију одређујемо преко својстава, ставова (теорема), и табеле парова. У табели постоје Лапласови трансформациони парови

In[4]:=
```
(s + a)/((s + a)^2 + b^2) == LT[InverseLaplaceTransform[(s + a)/((s + a)^2 + b^2), s, t]] // TraditionalForm
```

Out[4]//TraditionalForm=

$$\frac{a+s}{(a+s)^2+b^2} = \mathrm{LT}(e^{-a\,t}\cos(b\,t))$$

In[5]:=
```
b/((s + a)^2 + b^2) == LT[InverseLaplaceTransform[b/((s + a)^2 + b^2), s, t]] // TraditionalForm
```

Out[5]//TraditionalForm=

$$\frac{b}{(a+s)^2+b^2} = \mathrm{LT}(e^{-a\,t}\sin(b\,t))$$

Комплексну функцију преуређујемо на облик који одговара облику из табеле парова. Препознајемо коефицијенте из табеле. Овакав поступак се назива уклапање образаца (pattern matching).

$$\underline{H}(\underline{s}) = \frac{\underline{V}_3(\underline{s})}{\underline{U}_g(\underline{s})} = \frac{4\,\Omega\,\underline{s}}{3\,\underline{s}^2+2\,\Omega\,\underline{s}+3\,\Omega^2}, \quad \underline{H}(\underline{s}) = \frac{4}{3}\,\Omega\,\frac{\underline{s}}{\underline{s}^2+2\,\frac{1}{3}\,\Omega\,\underline{s}+\Omega^2}, \quad \underline{H}(\underline{s}) = \frac{4}{3}\,\Omega\,\frac{\underline{s}+\frac{1}{3}\,\Omega-\frac{1}{3}\,\Omega}{\left(\underline{s}+\frac{1}{3}\,\Omega\right)^2+\left(\frac{2\sqrt{2}}{3}\,\Omega\right)^2}$$

Функцију у временском домену одређујемо из Лапласовог трансформационог пара заменом коефицијената конкретним вредностима.

$$g(t) = \mathrm{LT}^{-1}(\underline{H}(\underline{s})) = \mathrm{LT}^{-1}\left(\frac{4}{3}\,\Omega\,\frac{\underline{s}+\frac{1}{3}\,\Omega-\frac{1}{3}\,\Omega}{\left(\underline{s}+\frac{1}{3}\,\Omega\right)^2+\left(\frac{2\sqrt{2}}{3}\,\Omega\right)^2}\right)$$

$$g(t) = \mathrm{LT}^{-1}\left(\frac{4}{3}\,\Omega\,\frac{\underline{s}+\frac{1}{3}\,\Omega}{\left(\underline{s}+\frac{1}{3}\,\Omega\right)^2+\left(\frac{2\sqrt{2}}{3}\,\Omega\right)^2}\right) - \mathrm{LT}^{-1}\left(\frac{4}{3\cdot 2\sqrt{2}}\,\Omega\,\frac{\frac{2\sqrt{2}}{3}\,\Omega}{\left(\underline{s}+\frac{1}{3}\,\Omega\right)^2+\left(\frac{2\sqrt{2}}{3}\,\Omega\right)^2}\right)$$

Додајемо генералисане функције и домен.

$$g(t) = \vartheta(t)\,\frac{4}{3}\,\Omega\,e^{\frac{-1}{3}\,\Omega\,t}\cos\left(\frac{2\sqrt{2}}{3}\,\Omega\,t\right) - \vartheta(t)\,\frac{\sqrt{2}}{3}\,\Omega\,e^{\frac{-1}{3}\,\Omega\,t}\sin\left(\frac{2\sqrt{2}}{3}\,\Omega\,t\right)$$

$$g(t) = \vartheta(t)\,\frac{1}{3}\,\Omega\,e^{\frac{-1}{3}\,\Omega\,t}\left(4\cos\left(\frac{2\sqrt{2}}{3}\,\Omega\,t\right) - \sqrt{2}\,\sin\left(\frac{2\sqrt{2}}{3}\,\Omega\,t\right)\right),$$

$$-\infty < t < +\infty$$

Одскочни одзив (индициона функција, step response, unit step response) је инверзна унилатерална Лапласова трансформација одговарајуће трансфер функције подељене са $\underline{s}$.

$$f(t) = \mathrm{LT}^{-1}\left(\frac{1}{\underline{s}}\,\underline{H}(\underline{s})\right) = \mathrm{LT}^{-1}\,\frac{4}{3}\,\Omega\,\frac{1}{\left(\underline{s}+\frac{1}{3}\,\Omega\right)^2+\left(\frac{2\sqrt{2}}{3}\,\Omega\right)^2},$$

$$f(t) = \mathrm{LT}^{-1}\,\sqrt{2}\,\frac{\frac{2\sqrt{2}}{3}\,\Omega}{\left(\underline{s}+\frac{1}{3}\,\Omega\right)^2+\left(\frac{2\sqrt{2}}{3}\,\Omega\right)^2},$$

Додајемо генералисане функције и домен.

$$f(t) = \vartheta(t)\,\sqrt{2}\,e^{\frac{-1}{3}\,\Omega\,t}\,\sin\!\left(\frac{2\sqrt{2}}{3}\,\Omega\,t\right),\quad -\infty < t < +\infty$$

Импулсни одзив (Гринова функција) је први извод по времену одскочног одзива (индиционе функције), $g(t) = \frac{d}{dt}\,f(t)$. Ово можемо искористити за алтернативно одређивање импулсног одзива преко одскочног одзива. Импулсни одзив се обележава и са $h(t)$, као словна асоцијација на $\underline{H}(\underline{s})$ и Лапласов трансформациони пар $(h(t),\,\underline{H}(\underline{s}))$.



Аутоматизована симболичка рачунарска анализа

```
In[6]:= $Assumptions = {R > 0, Ω > 0, ω ∈ Reals}
Out[6]= {R > 0, Ω > 0, ω ∈ ℝ}

In[7]:= okvir[x_] := x // Style[#, 24] & // Framed[#, FrameStyle → Cyan] & // TraditionalForm

In[8]:= okvirY[x_] := x // Style[#, 24] & // Framed[#, FrameStyle → Yellow] & // TraditionalForm

In[9]:= sZapis = {H → Style["H", {Underlined, Italic}][Style[s, Underlined]],
          Rule → Equal, s → Style[s, Underlined],
          Ug → Subscript[Style["U", {Underlined, Italic}], "g"][Style[s, Underlined]],
          V2 → Subscript[Style["V", {Underlined, Italic}], 2][Style[s, Underlined]],
          V3 → Subscript[Style["V", {Underlined, Italic}], 3][Style[s, Underlined]],
          V4 → Subscript[Style["V", {Underlined, Italic}], 4][Style[s, Underlined]]}
```

Out[9]= $\{H \to \underline{H}[\underline{s}],\ \mathrm{Rule} \to \mathrm{Equal},\ s \to \underline{s},\ \mathrm{Ug} \to \underline{U}_g[\underline{s}],\ \mathrm{V2} \to \underline{V}_2[\underline{s}],\ \mathrm{V3} \to \underline{V}_3[\underline{s}],\ \mathrm{V4} \to \underline{V}_4[\underline{s}]\}$

In[10]:=
```
zamena = {C1 → 1/(R Ω), C2 → 1/(R Ω), R1 → 3/2 R, R2 → R, R3 → R, R4 → 3/4 R, R7 → R, R8 → R}
```

Out[10]= $\left\{C1 \to \frac{1}{R\,\Omega},\ C2 \to \frac{1}{R\,\Omega},\ R1 \to \frac{3\,R}{2},\ R2 \to R,\ R3 \to R,\ R4 \to \frac{3\,R}{4},\ R7 \to R,\ R8 \to R\right\}$

Анализа помоћу унилатералне Лапласове трансформације

In[11]:=
```
jednacine =
 {-Ug/R4 + C1 s (-V2) + -V2/R1 + -V4/R3 == 0, -V2/R7 + -V3/R8 == 0, -V3/R2 + C2 s (-V4) == 0} /. zamena
```

Out[11]= $\left\{-\frac{4\,Ug}{3\,R} - \frac{2\,V2}{3\,R} - \frac{V4}{R} - \frac{s\,V2}{R\,\Omega} == 0,\ -\frac{V2}{R} - \frac{V3}{R} == 0,\ -\frac{V3}{R} - \frac{s\,V4}{R\,\Omega} == 0\right\}$

In[12]:=
```
Column[jednacine] /. sZapis // okvirY
```

Out[12]//TraditionalForm=

$$-\frac{4\,U_g(s)}{3\,R} - \frac{s\,V_2(s)}{R\,\Omega} - \frac{2\,V_2(s)}{3\,R} - \frac{V_4(s)}{R} = 0$$
$$-\frac{V_2(s)}{R} - \frac{V_3(s)}{R} = 0$$
$$-\frac{s\,V_4(s)}{R\,\Omega} - \frac{V_3(s)}{R} = 0$$

In[13]:=
```
odziv = Solve[jednacine, {V2, V3, V4}]
```

Out[13]= $\left\{\left\{V2 \to -\frac{4\,s\,Ug\,\Omega}{3\,s^2 + 2\,s\,\Omega + 3\,\Omega^2},\ V3 \to \frac{4\,s\,Ug\,\Omega}{3\,s^2 + 2\,s\,\Omega + 3\,\Omega^2},\ V4 \to -\frac{4\,Ug\,\Omega^2}{3\,s^2 + 2\,s\,\Omega + 3\,\Omega^2}\right\}\right\}$

In[14]:=
```
Column[odziv // First] /. sZapis // okvirY
```

Out[14]//TraditionalForm=

$$V_2(s) = -\frac{4\,s\,\Omega\,U_g(s)}{2\,s\,\Omega + 3\,s^2 + 3\,\Omega^2}$$
$$V_3(s) = \frac{4\,s\,\Omega\,U_g(s)}{2\,s\,\Omega + 3\,s^2 + 3\,\Omega^2}$$
$$V_4(s) = -\frac{4\,\Omega^2\,U_g(s)}{2\,s\,\Omega + 3\,s^2 + 3\,\Omega^2}$$

In[15]:=
```
Hs = V3/Ug /. First[odziv]
```

Out[15]= $\frac{4\,s\,\Omega}{3\,s^2 + 2\,s\,\Omega + 3\,\Omega^2}$

In[16]:=
```
H == Hs /. sZapis // okvir
```

Out[16]//TraditionalForm=

$$H(s) = \frac{4\,s\,\Omega}{2\,s\,\Omega + 3\,s^2 + 3\,\Omega^2}$$

In[17]:= `nuleHs = Solve[Numerator[Hs] == 0, s]`

Out[17]= `{{s → 0}}`

In[18]:= `nule == nuleHs /. sZapis // okvir`

Out[18]//TraditionalForm=

$$\boxed{\text{nule} = (\underline{s} = 0)}$$

In[19]:= `poloviHs = Solve[Denominator[Hs] == 0, s] // Simplify`

Out[19]= $\left\{\left\{s \to \frac{1}{3}\left(-1 - 2 i \sqrt{2}\right) \Omega\right\}, \left\{s \to \frac{1}{3}\left(-1 + 2 i \sqrt{2}\right) \Omega\right\}\right\}$

In[20]:= `polovi == poloviHs /. sZapis // okvir`

Out[20]//TraditionalForm=

$$\boxed{\text{polovi} = \begin{pmatrix} \underline{s} = \frac{1}{3}\left(-1 - 2 i \sqrt{2}\right) \Omega \\ \underline{s} = \frac{1}{3}\left(-1 + 2 i \sqrt{2}\right) \Omega \end{pmatrix}}$$

In[21]:= `Hω = Hs /. s → i ω`

Out[21]= $\frac{4 i \omega \Omega}{-3 \omega^2 + 2 i \omega \Omega + 3 \Omega^2}$

In[22]:= `Mω = ComplexExpand[Abs[Hω]] // Simplify`

Out[22]= $\frac{4 \Omega \,\text{Abs}[\omega]}{\sqrt{9 \omega^4 - 14 \omega^2 \Omega^2 + 9 \Omega^4}}$

In[23]:= `M[ω] == Mω // okvir`

Out[23]//TraditionalForm=

$$\boxed{M(\omega) = \frac{4 \Omega |\omega|}{\sqrt{9 \omega^4 - 14 \omega^2 \Omega^2 + 9 \Omega^4}}}$$

In[24]:= `{Mmax, ωmaxRule} = Maximize[Mω, ω] // FullSimplify`

Out[24]= `{2, {ω → -Ω}}`

In[25]:= `Mref = Mmax`

Out[25]= `2`

In[26]:= `ω12 = Select[ω /. Solve[Mω^2 == 1/2 Mref^2, ω], Simplify[Positive[#]] &] // Factor`

Out[26]= $\left\{\frac{1}{3}\left(-1 + \sqrt{10}\right) \Omega, \frac{1}{3}\left(1 + \sqrt{10}\right) \Omega\right\}$

In[27]:= `ω1 = Min[ω12] // Simplify`

Out[27]= $\frac{1}{3}\left(-1 + \sqrt{10}\right) \Omega$

In[28]:= ω2 = Max[ω12] // Simplify

Out[28]= $\frac{1}{3}\left(1+\sqrt{10}\right)\Omega$

In[29]:= Bω3dB = ω2 - ω1 // Simplify

Out[29]= $\frac{2\,\Omega}{3}$

In[30]:= Row[{ω_1 == ω1, ",   ", ω_2 == ω2, ",   ", $B_{\omega 3dB}$ == Bω3dB}] // okvir

Out[30]//TraditionalForm=

$$\boxed{\omega_1 = \frac{1}{3}\left(\sqrt{10}-1\right)\Omega, \quad \omega_2 = \frac{1}{3}\left(1+\sqrt{10}\right)\Omega, \quad B_{\omega 3\mathrm{dB}} = \frac{2\,\Omega}{3}}$$

In[31]:= gt = InverseLaplaceTransform[Hs, s, t] // Expand // Collect[#, ⅇ-] &

Out[31]= $e^{-\frac{t\Omega}{3}}\left(\frac{4}{3}\Omega\,\mathrm{Cos}\left[\frac{2}{3}\sqrt{2}\,t\,\Omega\right]-\frac{1}{3}\sqrt{2}\,\Omega\,\mathrm{Sin}\left[\frac{2}{3}\sqrt{2}\,t\,\Omega\right]\right)$

In[32]:= Row[{g[t] == ϑ[t] gt, ",   ", -∞ < t < ∞}] // okvirY

Out[32]//TraditionalForm=

$$\boxed{g(t) = e^{-\frac{t\Omega}{3}}\vartheta(t)\left(\frac{4}{3}\Omega\cos\left(\frac{2}{3}\sqrt{2}\,t\,\Omega\right)-\frac{1}{3}\sqrt{2}\,\Omega\sin\left(\frac{2}{3}\sqrt{2}\,t\,\Omega\right)\right), \quad -\infty < t < \infty}$$

In[33]:= Plot[$\left(\frac{gt}{\Omega} \text{ /. } t \to \frac{x}{\Omega}\right)$ // Evaluate, {x, 0, 15}, AxesLabel → {Ω t, g[t]},
PlotRange → All, Epilog → Text["(0,0)", {0.65, -0.09}]]

Out[33]=



In[34]:= ft = InverseLaplaceTransform[$\frac{1}{s}$ Hs, s, t] // Expand // Collect[#, ⅇ-] &

Out[34]= $\sqrt{2}\,e^{-\frac{t\Omega}{3}}\,\mathrm{Sin}\left[\frac{2}{3}\sqrt{2}\,t\,\Omega\right]$

In[35]:=
```
Row[{f[t] == ϑ[t] ft, ",    ", -∞ < t < ∞}] // okvir
```

Out[35]//TraditionalForm=

$$\boxed{f(t) = \sqrt{2}\, e^{-\frac{t\,\Omega}{3}}\, \vartheta(t) \sin\left(\frac{2}{3}\sqrt{2}\, t\,\Omega\right), \quad -\infty < t < \infty}$$

In[36]:=
```
Plot[(ft /. t → x/Ω) // Evaluate, {x, 0, 15},
  AxesLabel → {Ω t, f[t]}, Epilog → Text["(0,0)", {0.65, -0.056}]]
```

Out[36]=



Mathematica

In[37]:=
```
$Version
```

Out[37]=
```
11.2.0 for Microsoft Windows (64-bit) (September 11, 2017)
```

In[38]:=
```
DateString[]
```

Out[38]=
```
Thu 19 Dec 2019 22:32:40
```

```
(%i1) load("C:\\SALECx\\SALECx.mac") $
      Dejan Tosic, SALECx 2019 v1.0
      Symbolic Analysis of Linear Electric Circuits with Maxima

(%i2) SALECxPrint: true $

(%i3) TT_shema: [
       ["V", "Ug", 1, 0, Ug],
       ["R", "R4", 1, 5, R4],
       ["OpAmp", "OpAmp1", [0,5], 2],
       ["OpAmp", "OpAmp2", [0,6], 3],
       ["OpAmp", "OpAmp3", [0,7], 4],
       ["C", "C1", 5, 2, C1],
       ["C", "C2", 7, 4, C2],
       ["R", "R1", 5, 2, R1],
       ["R", "R3", 5, 4, R3],
       ["R", "R7", 2, 6, R7],
       ["R", "R8", 6, 3, R8],
       ["R", "R2", 3, 7, R2],
       ["R", "R9", 4, 0, R9]
      ],
      [
       C1=1/(R·Omega), C2=1/(R·Omega),
       R1=3·R/2, R2=R, R3=R, R4=3·R/4, R7=R, R8=R
      ]$
```

(%i4) TT_response: SALECx(TT_shema);

Symbolic Analysis of Linear Electric Circuits with Maxima
SALECx version 1.0, Prof. Dr. Dejan Tošić, tosic@etf.rs
Number of nodes excluding 0 node: 7

Electric circuit specification: $[[V,Ug,1,0,Ug],[R,R4,1,5,\frac{3R}{4}],[OpAmp,OpAmp1,[0,5],2],[OpAmp,OpAmp2,[0,6],3],[OpAmp,OpAmp3,[0,7],4],[C,C1,5,2,\frac{1}{\Omega R}],[C,C2,7,4,\frac{1}{\Omega R}],[R,R1,5,2,\frac{3R}{2}],[R,R3,5,4,R],[R,R7,2,6,R],[R,R8,6,3,R],[R,R2,3,7,R],[R,R9,4,0,R9]]$

Supported element: [**true**, **true**, **true**, **true**, **true**, **true**, **true**, **true**, **true**, **true**, **true**, **true**, **true**]

Element values: $[Ug,\frac{3R}{4},$ **false**, **false**, **false**, $\frac{1}{\Omega R},\frac{1}{\Omega R},\frac{3R}{2},R,R,R,R,R9]$

Initial conditions: [**false**, **false**, **false**, **false**, **false**, **false**, **false**, **false**, **false**, **false**, **false**, **false**, **false**]

MNA equations: $$\left[\frac{4(V_1-V_5)}{3R}+I_{Ug}=0,\frac{(V_2-V_5)s}{\Omega R}+\frac{V_2-V_6}{R}+\frac{2(V_2-V_5)}{3R}+I_{OpAmp1}=0,\frac{V_3-V_7}{R}+\frac{V_3-V_6}{R}+I_{OpAmp2}=0,\frac{(V_4-V_7)s}{\Omega R}+\frac{V_4}{R9}+\frac{V_4-V_5}{R}+I_{OpAmp3}=0,\frac{(V_5-V_2)s}{\Omega R}+\frac{V_5-V_4}{R}+\frac{2(V_5-V_2)}{3R}+\frac{4(V_5-V_1)}{3R}=0,\frac{V_6-V_3}{R}+\frac{V_6-V_2}{R}=0,\frac{(V_7-V_4)s}{\Omega R}+\frac{V_7-V_3}{R}=0,-V_7=0,-V_6=0,-V_5=0,V_1=Ug\right]$$

MNA variables: $[V_1,V_2,V_3,V_4,V_5,V_6,V_7,I_{OpAmp3},I_{OpAmp2},I_{OpAmp1},I_{Ug}]$

(TT_response) $$\left[V_1=Ug,V_2=-\frac{4\,\Omega\,Ug\,s}{3s^2+2\,\Omega\,s+3\,\Omega^2},V_3=\frac{4\,\Omega\,Ug\,s}{3s^2+2\,\Omega\,s+3\,\Omega^2},V_4=-\frac{4\,\Omega^2\,Ug}{3s^2+2\,\Omega\,s+3\,\Omega^2},V_5=0,V_6=0,V_7=0,I_{OpAmp3}=\frac{4\,\Omega\,R9\,Ug\,s+4\,\Omega^2\,R9\,Ug+4\,\Omega^2\,R\,Ug}{R(3\,R9\,s^2+2\,\Omega\,R9\,s+3\,\Omega^2\,R9)},I_{OpAmp2}=-\frac{8\,\Omega\,Ug\,s}{R(3s^2+2\,\Omega\,s+3\,\Omega^2)},I_{OpAmp1}=\frac{12\,Ug\,s^2+20\,\Omega\,Ug\,s}{R(9s^2+6\,\Omega\,s+9\,\Omega^2)},I_{Ug}=-\frac{4\,Ug}{3R}\right]$$

(%i5) Hs: V[3]/Ug, TT_response;

(Hs) $$\frac{4\,\Omega\,s}{3s^2+2\,\Omega\,s+3\,\Omega^2}$$

(%i6) nuleHs: solve(num(Hs)=0,s);

(nuleHs) $[s=0]$

(%i7) poloviHs: solve(denom(Hs)=0,s);

(poloviHs) $[s=-\frac{(2^{3/2}\%i+1)\,\Omega}{3},s=\frac{(2^{3/2}\%i-1)\,\Omega}{3}]$

(%i8) Homega: Hs, s=%i·omega;

(Homega) $\frac{4\,\%i\,\Omega\,\omega}{-3\,\omega^2+2\,\%i\,\Omega\,\omega+3\,\Omega^2}$

(%i9) assume(omega>0,Omega>0);

(%o9) $[\omega>0,\Omega>0]$

(%i10) Momega: cabs(Homega);

(Momega) $\frac{4\,\Omega\,\omega}{\sqrt{(3\,\Omega^2-3\,\omega^2)^2+4\,\Omega^2\,\omega^2}}$

(%i11) gt: ilt(Hs,s,t);

(gt) $\frac{\%e^{-\frac{\Omega\,t}{3}}\left(4\,\Omega\cos\left(\frac{2^{3/2}\,\Omega\,t}{3}\right)-\sqrt{2}\,\Omega\sin\left(\frac{2^{3/2}\,\Omega\,t}{3}\right)\right)}{3}$

(%i12) ft: ilt((1/s)·Hs,s,t);

(ft) $\sqrt{2}\,\%e^{-\frac{\Omega\,t}{3}}\sin\left(\frac{2^{3/2}\,\Omega\,t}{3}\right)$

(%i13) Fomega: 1/Momega^2;

(Fomega) $\frac{(3\,\Omega^2-3\,\omega^2)^2+4\,\Omega^2\,\omega^2}{16\,\Omega^2\,\omega^2}$

(%i14) omegaSolve: solve(diff(Fomega,omega) = 0,omega);

(omegaSolve) $[\omega=\%i\,\Omega,\omega=-\Omega,\omega=-\%i\,\Omega,\omega=\Omega]$

(%i15) omegaMax: omega, last(omegaSolve);

(omegaMax) Ω

(%i16) omegaRef: omegaMax;

(omegaRef) Ω

(%i17) Mref: Momega, omega=omegaRef;

(Mref) 2

(%i18) omega12: solve(Momega^2 = (1/2)·Mref^2, omega), ratsimp;

(omega12) $[\omega=-\frac{\sqrt{2\sqrt{10}+11}\,\Omega}{3},\omega=\frac{\sqrt{2\sqrt{10}+11}\,\Omega}{3},\omega=-\frac{\sqrt{11-2\sqrt{10}}\,\Omega}{3},\omega=\frac{\sqrt{11-2\sqrt{10}}\,\Omega}{3}]$

(%i19) omega1: omega, last(omega12);

$$\text{(omega1)}\quad \frac{\sqrt{11-2\sqrt{10}}\,\Omega}{3}$$

(%i20) omega2: omega, second(omega12);

$$\text{(omega2)}\quad \frac{\sqrt{2\sqrt{10}+11}\,\Omega}{3}$$

(%i21) Bomega3dB: omega2 − omega1, ratsim;

$$\text{(Bomega3dB)}\quad \frac{\sqrt{2\sqrt{10}+11}\,\Omega}{3}-\frac{\sqrt{11-2\sqrt{10}}\,\Omega}{3}$$

(%i22) float(Bomega3dB);

(%o22) $0.6666666666666665\ \Omega$